CL643 : Assignment 4

(solving Assignment-3 problems in GAMS)

Name : Rohit Priyanka Nandwani
Roll :180121035
Date :16th October, 2021

Problem Statement :

- Solve the ten problems (Q1 to Q5) in GAMS (using NEOS).

****PLEASE NOTE****

All the related files [.gms, .log, .lst., .svg, .txt files] can be found on the following github repository ::

https://github.com/nandwani-rohit/CL643Assignment4

** Also, I have unable to complete the Multi-Objective NLP and Multi-Objective MINLP Problems within the deadline.. However, I will try to complete those too, and their respective code files can also be found in the above repository.

The updated pdf file (with MONLP+MOMINLP), can be found at :
https://nandwani-rohit.github.io/CL643Assignment4/180121035_Assignment_4.pdf

---

1) Linear Programming Problem

Feasible

$$Minimize \;\; f(x_1, x_2, x_3) \; = \; 3\,x_1 - 4\,x_2 + 53\,x_3$$

subject to the constraints :

$$x_1 \geqslant 0$$
$$x_2 \geqslant 0$$
$$x_3 \geqslant 0$$
$$\pi\,x_1 - x_2 + 20\,x_3 \geqslant 43$$
$$7\,x_1 + x_2 = \frac{7}{\pi} + \log_e(10) \;\approx\; 4.53075$$

This problem is feasible, and infact has the optimal solution :

$$\mathbf{x} = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 0 \\ \dfrac{7}{\pi} + \log_e(10) \\ \dfrac{7}{20\pi} + \dfrac{43}{20} + \dfrac{\log_e(10)}{20} \end{pmatrix} \approx \begin{pmatrix} 0 \\ 4.53075 \\ 2.37654 \end{pmatrix}$$

≈ GAMS CODE

```
Variables
    x1       x1
    x2       x2
    x3       x3
    f        Objective Function ;

Equations
    e1       Objective Function definition
    e2       Constraint 1
    e3       Constraint 2 ;

e1..     f =e= 3*x1 - 4*x2 + 53*x3;
e2..     pi*x1 - x2 + 20*x3 =g= 43;
e3..     7*x1 + x2 - 7/pi - log(10) =e= 0;

* Bounding the decision variables
x1.lo = 0;
x2.lo = 0;
x3.lo = 0;

Model q1_feasible /all/;

* Specifying the LP Solver
option LP = CPLEX;

Solve q1_feasible using LP minimizing f;

option decimals = 5;

display x1.l, x2.l, x3.l, f.l;
```

≈ **OUTPUT**

```
VARIABLE x1.L                  =       0.00000  x1
VARIABLE x2.L                  =       4.53075  x2
VARIABLE x3.L                  =       2.37654  x3
VARIABLE f.L                   =     107.83348  Objective Function
```

Infeasible

$$\textit{Minimize}\ \ f(x_1, x_2, x_3) = 3\,x_1 - 4\,x_2 + 53\,x_3$$

subject to the constraints :

$$x_1 \geqslant 0$$

$$x_2 \geqslant 0$$
$$x_3 \geqslant 0$$
$$\pi\, x_1 + x_2 - 20\, x_3 \geqslant 43$$
$$7\, x_1 + x_2 = \frac{7}{\pi} + \log_e(10) \;\approx\; 4.53075$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, optimal) solution.

$\mathbf{x}$ *does not exist.*

≈ GAMS CODE

```
Variables
          x1        x1
          x2        x2
          x3        x3
          f         Objective Function ;

Equations
          e1        Objective Function definition
          e2        Constraint 1
          e3        Constraint 2 ;

e1..      f =e= 3*x1 - 4*x2 + 53*x3;
e2..      pi*x1 + x2 - 20*x3 =g= 43;
e3..      7*x1 + x2 - 7/pi - log(10) =e= 0;

* Bounding the decision variables
x1.lo = 0;
x2.lo = 0;
x3.lo = 0;

Model q1_infeasible /all/;

* Specifying the LP Solver
option LP = CPLEX;

Solve q1_infeasible using LP minimizing f;

option decimals = 5;

display x1.l, x2.l, x3.l, f.l;
```

≈ OUTPUT

```
                    S O L V E      S U M M A R Y

     MODEL    q1_infeasible         OBJECTIVE   f
     TYPE     LP                    DIRECTION   MINIMIZE
     SOLVER   CPLEX                 FROM LINE    26

**** SOLVER STATUS      1 Normal Completion
**** MODEL STATUS       4 Infeasible
**** OBJECTIVE VALUE                 2.4043

 RESOURCE USAGE, LIMIT              0.047 10000000000.000
 ITERATION COUNT, LIMIT             0     2147483647

Model has been proven infeasible

                       LOWER          LEVEL          UPPER         MARGINAL

---- EQU e1              .              .              .             1.0000
---- EQU e2            43.0000         4.5308        +INF            0.0625  INFES
---- EQU e3             4.5308         4.5308         4.5308        -0.0625

  e1  Objective Function definition
  e2  Constraint 1
  e3  Constraint 2


 VARIABLE x1.L                    =        0.00000  x1
 VARIABLE x2.L                    =        4.53075  x2
 VARIABLE x3.L                    =        0.00000  x3
 VARIABLE f.L                     =        2.40433  Objective Function
```

---

2) <u>*Non-Linear Programming Problem*</u>

<u>Feasible</u>

$$Maximize \;\; f(x_1, x_2, x_3) \;=\; x_3 - \left(x_2 + \frac{x_1}{2}\right)^4 \exp\left(x_2^2 - x_1^2\right)$$

subject to the constraints :

$$-3\,x_1^2 + x_3 - 0.4 > 0$$
$$x_3 - 3.9 < 0$$
$$x_1 \cdot x_2 - x_3 = 0$$

This problem is feasible, and infact has the optimal solution :

$$\mathbf{x} \;=\; \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} \;=\; \begin{pmatrix} 0.364008 \\ 2.1909 \\ 0.797506 \end{pmatrix}$$

≈ <u>GAMS CODE</u>

```
Variables
         x1      x1
         x2      x2
         x3      x3
         f       Objective Function ;

Scalar
         tolerance  strict inequalities within a tolerance  /1e-8/;

Equations
         e1      Objective Function definition
         e2      Constraint 1
         e3      Constraint 2
         e4      Constraint 3 ;

e1..     f =e= x3 - power(x2+x1/2,4) * exp(sqr(x2)-sqr(x1));
e2..     -3*sqr(x1) + x3 - 0.4 =g= tolerance;
e3..     x3 - 3.9 =l= -tolerance;
e4..     x1*x2 - x3  =e= tolerance;

* Starting position in the feasible region
x1.l = 1;
x2.l = 1;
x3.l = 1;

Model q2_feasible /all/;

* Specifying the NLP Solver
option NLP = CONOPT;

Solve q2_feasible using NLP maximizing f;

option decimals = 6;

display x1.l, x2.l, x3.l, f.l;
```

≈ <u>OUTPUT</u>

```
VARIABLE x1.L                 =     0.364008  x1
VARIABLE x2.L                 =     2.190901  x2
VARIABLE x3.L                 =     0.797506  x3
VARIABLE f.L                  =  -3.37371E+3  Objective Function
```

<u>Infeasible</u>

$$Maximize\;\; f(x_1,x_2,x_3) \;=\; x_1^2 + x_2^2 - 3\,x_1 \cdot \sin(x_2) \cdot x_3^2$$

subject to the constraints :

$$-x_1^2 + 3\,x_1 + x_2 - x_3 - 5.25 > 0$$
$$-3\sin(x_1 - 1) + x_2 = 0$$

$$x_3 > 0$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, optimal) solution.

$$\mathbf{x} \text{ does not exist.}$$

≈ GAMS CODE

```
Variables
         x1       x1
         x2       x2
         x3       x3
         f        Objective Function ;

Scalar
         tolerance  strict inequalities within a tolerance  /1e-8/;

Equations
         e1       Objective Function definition
         e2       Constraint 1
         e3       Constraint 2 ;
*         e4       Constraint 3;

e1..     f =e= sqr(x1) + sqr(x2) - 3*x1*sin(x2)*sqr(x3);
e2..     -sqr(x1) + 3*x1 + x2 - x3 - 5.25 =g= tolerance;
e3..     -3*sin(x1-1) + x2 =e= tolerance;
*e4..     x3 =g= tolerance;

*Equivalently e4 can simply be written as lower bound constraint
x3.lo = 0;

* Starting position in the feasible region
x1.l = 1;
x2.l = 1;
x3.l = 1;

Model q2_infeasible /all/;

* Specifying the NLP Solver
option NLP = CONOPT;

Solve q2_infeasible using NLP maximizing f;

option decimals = 6;

display x1.l, x2.l, x3.l, f.l;
```

≈ OUTPUT

```
                S O L V E      S U M M A R Y

     MODEL    q2_infeasible       OBJECTIVE  f
     TYPE     NLP                 DIRECTION  MAXIMIZE
     SOLVER   CONOPT              FROM LINE  34

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      5 Locally Infeasible
**** OBJECTIVE VALUE               11.9459

 RESOURCE USAGE, LIMIT          0.047 10000000000.000
 ITERATION COUNT, LIMIT        17     2147483647
 EVALUATION ERRORS              0              0


** Infeasible solution.

                           LOWER          LEVEL          UPPER         MARGINAL
---- EQU e1                  .              .              .              EPS
---- EQU e2                5.2500         4.5664        +INF            0.2500  INFES
---- EQU e3            1.0000000E-8   1.0000000E-8   1.0000000E-8      -0.2500

  e1  Objective Function definition
  e2  Constraint  1
  e3  Constraint  2


VARIABLE x1.L                     =      2.134237  x1
VARIABLE x2.L                     =      2.718635  x2
VARIABLE x3.L                     =      0.000000  x3
VARIABLE f.L                      =     11.945942  Objective Function
```

---

3) *Mixed Integer Linear Programming Problem*

Feasible

$$\textit{Minimize}\ \ f(x_1,x_2,x_3) \ = \ 5\,x_1 - x_2 + 10\,x_3$$

subject to the constraints :

$$x_3 \ \in \ \{0,1,2,\ ...,20\} \ = \ \mathbb{Z}[0,20]$$
$$x_1,x_2 \ \geqslant \ 0$$

$$15\,x_1 - x_3 \geqslant 3$$
$$x_2 + 9\,x_1 \leqslant 6$$
$$-9\,x_1 + 4x_2 = 15$$

This problem is feasible, and infact has the optimal solution :

$$\mathbf{x} \ = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 1/5 \\ 21/5 \\ 0 \end{pmatrix} = \begin{pmatrix} 0.2 \\ 4.2 \\ 0 \end{pmatrix}$$

≈ GAMS CODE

```
Variables
          x1        x1
          x2        x2
          x3        x3
          f         Objective Function ;

Integer Variable
          x3        x3;

Nonnegative Variables
          x1        x1
          x2        x2;

Equations
          e1        Objective Function definition
          e2        Constraint 1
          e3        Constraint 2
          e4        Constraint 3 ;

e1..      f =e= 5*x1 - x2 + 10*x3;
e2..      15*x1 - x3 =g= 3;
e3..      x2 + 9*x1 =l= 6;
e4..      -9*x1 + 4*x2 =e= 15;

* Bounding the decision variables
x3.up = 20;

Model q3_feasible /all/;

* Specifying the MIP Solver
option MIP = CPLEX;

Solve q3_feasible using MIP minimizing f;

option decimals = 5;

display x1.l, x2.l, x3.l, f.l;
```

≈ OUTPUT

```
VARIABLE x1.L                =      0.20000  x1
VARIABLE x2.L                =      4.20000  x2
VARIABLE x3.L                =      0.00000  x3
VARIABLE f.L                 =     -3.20000  Objective Function
```

Infeasible

$$Maximize \;\; f(x_1, x_2, x_3) \; = \; x_1 + x_2 + 15\, x_3$$

subject to the constraints :

$$x_1,\ x_3 \;\in\; \mathbb{Z}[0, \infty)$$
$$x_2 \geqslant 0$$

$$2\,x_1 - x_2 \geqslant 6$$
$$x_2 + x_3 \geqslant 4$$
$$x_1 \leqslant 3.9$$
$$x_1 + 2\,x_2 + 3\,x_3 = 10$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, optimal) solution.

$$\mathbf{x} \;\; does\ not\ exist.$$

≈ GAMS CODE

```
Variables
        x1      x1
        x2      x2
        x3      x3
        f       Objective Function ;

Integer Variable
        x1      x1
        x3      x3;

Nonnegative Variables
        x2      x2;

Equations
        e1      Objective Function definition
        e2      Constraint 1
        e3      Constraint 2
        e4      Constraint 3 ;

e1..    f =e= x1 + x2 + 15*x3;
e2..    2*x1 - x2 =g= 6;
e3..    x2 + x3 =g= 4;
e4..    x1 + 2*x2 + 3*x3 =e= 10;

* Bounding the decision variables
* because x1 <= 3.9 is equivalent to x1 <= floor(3.9) = 3
x1.up = 3;

Model q3_infeasible /all/;

* Specifying the MIP Solver
option MIP = CPLEX;

Solve q3_infeasible using MIP minimizing f;

option decimals = 5;

display x1.l, x2.l, x3.l, f.l;
```

≈ **OUTPUT**

```
                    S O L V E      S U M M A R Y

     MODEL   q3_infeasible       OBJECTIVE  f
     TYPE    MIP                 DIRECTION  MINIMIZE
     SOLVER  CPLEX               FROM LINE  35

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      10 Integer Infeasible
**** OBJECTIVE VALUE                 NA

 RESOURCE USAGE, LIMIT          0.031 10000000000.000
 ITERATION COUNT, LIMIT         0    2147483647


--- MIP status (103): integer infeasible.
--- Cplex Time: 0.00sec (det. 0.00 ticks)

--- Problem is integer infeasible

No solution returned
GAMS 36.2.0  r433180e

VARIABLE x1.L                  =      0.00000  x1
VARIABLE x2.L                  =      0.00000  x2
VARIABLE x3.L                  =      0.00000  x3
VARIABLE f.L                   =      0.00000  Objective Function
```

---

4) Mixed Integer Non-Linear Programming Problem

Feasible

$$^{**}\mathit{Maximize}\ \ f(x_1, x_2, x_3) \ = \ x_1^3 \cdot x_2^3 - x_1 \cdot x_2^2 \cdot \sin(x_3)$$

subject to the constraints :

$$x_2,\ x_3 \ \in \ \mathbb{Z}[0, 100]$$
$$x_1 \geqslant 0$$
$$x_1 \leqslant 100$$
$$-4\, x_1^3 + x_2 \geqslant 0$$
$$x_2 - x_3 - 3 \leqslant 0$$
$$13\, x_1 - 1.5\, x_2 \ + 1.98\, x_3 \ = 84.03226$$

This problem is feasible, and infact has the optimal solution :

$$\mathbf{x} \ = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} \ = \ \begin{pmatrix} 2.92402 \\ 100 \\ 99 \end{pmatrix}$$

≈ GAMS CODE

```
Variables
        x1      x1
        x2      x2
        x3      x3
        f       Objective Function ;

Integer Variable
        x2      x2
        x3      x3;

Nonnegative Variables
        x1      x1;

Equations
        e1      Objective Function definition
        e2      Upper Bound on x1
        e3      Upper Bound on x2
        e4      Upper Bound on x3
        e5      Constraint 1
        e6      Constraint 2
        e7      Constraint 3 ;

e1..    f =e= power(x1,3)*power(x2,3) - x1*sqr(x2)*sin(x3);
e2..    x1 =l= 100;
e3..    x2 =l= 100;
e4..    x3 =l= 100;
e5..    -4*power(x1,3) + x2 =g= 0;
e6..    x2 - x3 - 3 =l= 0;
e7..    13*x1 - 1.5*x2 + 1.98*x3 =e= 84.03226;

Model q4_feasible /all/;

* Specifying the MINLP Solver
option MINLP = LINDO;

Solve q4_feasible using MINLP maximizing f;

option decimals = 5;

display x1.l, x2.l, x3.l, f.l;
```

≈ OUTPUT

```
VARIABLE x1.L                    =      2.92402  x1
VARIABLE x2.L                    =    100.00000  x2
VARIABLE x3.L                    =     99.00000  x3
VARIABLE f.L                     =  2.502922E+7  Objective Function
```

Infeasible

$$Maximize\;\; f(x_1, x_2, x_3) \;=\; x_1^3 \cdot x_2^3 \;-\; x_1 \cdot x_2^2 \cdot \cos(x_3)$$

subject to the constraints :

$$x_2,\; x_3 \;\in\; \mathbb{Z}[0, 100]$$

$$x_1 \geqslant -\pi/2$$

$$x_1 \leqslant \pi$$

$$\cos(x_1)^2 - x_2 + x_3 \geqslant 0$$

$$2 \cdot \cos(4)^2 \cdot x_1 - x_2 - 7 \cdot \cos(4)^2 = 0$$

or

$$(0.8545\; x_1 - x_2 - 2.99075 = 0)$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, optimal) solution.

x *does not exist.*

≈ GAMS CODE

```
Variables
          x1       x1
          x2       x2
          x3       x3
          f        Objective Function ;

Integer Variable
          x2       x2
          x3       x3;

Nonnegative Variables
          x1       x1;

Equations
          e1       Objective Function definition
          e2       Lower Bound on x1
          e3       Upper Bound on x1
          e4       Upper Bound on x2
          e5       Upper Bound on x3
          e6       Constraint 1
          e7       Constraint 2 ;

e1..      f =e= power(x1,3)*power(x2,3) - x1*sqr(x2)*cos(x3);
e2..      x1 =g= -pi/2;
e3..      x1 =l= pi;
e4..      x2 =l= 100;
e5..      x3 =l= 100;
e6..      sqr(cos(x1)) - x2 + x3 =g= 0;
e7..      2*sqr(cos(4))*x1 - x2 - 7*sqr(cos(4)) =e= 0;

Model q4_infeasible /all/;

* Specifying the MINLP Solver
option MINLP = LINDO;

Solve q4_infeasible using MINLP maximizing f;

option decimals = 5;

display x1.l, x2.l, x3.l, f.l;
```

≈ **OUTPUT**

```
                    S O L V E      S U M M A R Y

     MODEL   q4_infeasible       OBJECTIVE  f
     TYPE    MINLP               DIRECTION  MAXIMIZE
     SOLVER  LINDO               FROM LINE  37

**** SOLVER STATUS     1 Normal Completion
**** MODEL STATUS      19 Infeasible - No Solution
**** OBJECTIVE VALUE                0.0000

 RESOURCE USAGE, LIMIT          0.032 10000000000.000
 ITERATION COUNT, LIMIT         0     2147483647
 EVALUATION ERRORS             NA              0

--- The model is infeasible.

No solution returned
GAMS 36.2.0  r433180e

VARIABLE x1.L                =       0.00000  x1
VARIABLE x2.L                =       0.00000  x2
VARIABLE x3.L                =       0.00000  x3
VARIABLE f.L                 =       0.00000  Objective Function
```

---

5) *Multi-Objective versions of all above mentioned-problems*

(I) Multi-Objective Linear Programming Problem

Feasible

$$Minimize\ \ f(x_1,x_2,x_3)\ =\ 3\,x_1-4\,x_2+53\,x_3$$
$$and$$
$$Minimize\ \ g(x_1,x_2,x_3)\ =\ -10\,x_1\ +\ x_2-34\,x_3$$

subject to the constraints :

$$x_1 \geqslant 0$$
$$x_2 \geqslant 0$$
$$x_3 \geqslant 0$$
$$\pi\,x_1-x_2+20\,x_3 \geqslant 43$$
$$7\,x_1+x_2=\frac{7}{\pi}+\log_e(10)\ \approx\ 4.53075$$

This problem is feasible. For example, it has the following feasible solution:

$$\mathbf{x} = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 0 \\ \dfrac{7}{\pi} + \log_e(10) \\ \dfrac{7}{20\pi} + \dfrac{43}{20} + \dfrac{\log_e(10)}{20} \end{pmatrix} \approx \begin{pmatrix} 0 \\ 4.53075 \\ 2.37654 \end{pmatrix}$$

≈ GAMS CODE

```
$eolCom //

// Problem Definition....

Set
   I 'constraints'         / i1* i2 /
   J 'decision variables'  / j1*j3 /
   K 'objective functions' / k1* k2 /;

Parameter
   dir(k)  'direction of the objective functions 1 for max and -1 for min'   / k1 -1, k2 -1 /
   b(I)    'RHS of the constraints'  / i1 43, i2 4.530754296 /;

Table c(K,J)  'matrix of objective function coefficients C'
        j1  j2  j3
    k1  3   -4  53
    k2  -10 1   -34;

Table a(I,J)  'matrix of constraint coefficients A'
        j1               j2  j3
    i1  3.141592654      -1  20
    i2  7                1   0;

Variable
   Z(K)  'objective function variables'
   X(J)  'decision variables' ;

NonNegative Variable X;

Equation
   objfun(K)  'objective functions'
   con(I)     'constraints';

objfun(K).. sum(J, c(K,J)*X(J))  =e= Z(K);

con(I)..    sum(J, a(I,J)*X(J))  =l= b(I);

Model q5_1_feasible / all /;

// Code for improved epsilon-constraint method
// (AUGMENCON-2) BEGINS here...

Set
   k1(k)  'the first element of k'
   km1(k) 'all but the first elements of k'
   kk(k)  'active objective function in constraint allobj' ;

k1(k)$(ord(k) = 1) = yes;
km1(k)  = yes;
km1(k1) =  no;

Parameter
   rhs(k)     'right hand side of the constrained obj functions in eps-constraint'
   maxobj(k)  'maximum value from the payoff table'
   minobj(k)  'minimum value from the payoff table'
   numk(k)    'ordinal value of k starting with 1' ;

Scalar
   iter         'total number of iterations'
   infeas       'total number of infeasibilities'
   elapsed_time 'elapsed time for payoff and e-sonstraint'
   start        'start time'
   finish       'finish time';

Variable
   a_objval  'auxiliary variable for the objective function'
   obj       'auxiliary variable during the construction of the payoff table'
   sl(k)     'slack or surplus variables for the eps-constraints'  ;

Positive Variable sl;

Equation
   con_obj(k)  'constrained objective functions'
   augm_obj    'augmented objective function to avoid weakly efficient solutions'
```

```
   allobj      'all the objective functions in one expression' ;

con_obj(km1).. z(km1) - dir(km1)*sl(km1)  =e= rhs(km1);

* We optimize the first objective function and put the others as constraints
* the second term is for avoiding weakly efficient points

augm_obj..
   a_objval =e= sum(k1,dir(k1)*z(k1))
         + 1e-3*sum(km1,power(10,-(numk(km1) - 1))*sl(km1)/(maxobj(km1) - minobj(km1)));

allobj.. sum(kk, dir(kk)*z(kk))  =e= obj;

Model
   mod_payoff    / q5_1_feasible, allobj             /
   mod_epsmethod / q5_1_feasible, con_obj, augm_obj /;

Parameter payoff(k,k)  'payoff tables entries' ;

Alias (k,kp);

option optCr = 0, limRow = 0, limCol = 0, solPrint = off, solveLink = %solveLink.LoadLibrary%;

* Generate payoff table applying lexicographic optimization
loop(kp,
   kk(kp) = yes;
   repeat
      solve mod_payoff using mip maximizing obj;
      payoff(kp,kk) = z.l(kk);
      z.fx(kk) = z.l(kk);  // freeze the value of the last objective optimized
      kk(k++1) = kk(k);    // cycle through the objective functions
   until kk(kp);
   kk(kp) = no;
*  release the fixed values of the objective functions for the new iteration
   z.up(k) =   inf;
   z.lo(k) = -inf;
);
if(mod_payoff.modelStat <> %modelStat.optimal% ,
   abort 'no optimal solution for mod_payoff' ;);

File fx / q5_1_feasible_solutions.txt /;
put  fx ' PAYOFF TABLE'/;
loop(kp,
   loop(k, put payoff(kp,k):12:2;);
   put /;
);

minobj(k) = smin(kp,payoff(kp,k));
maxobj(k) = smax(kp,payoff(kp,k));

* for the current problem, gridpoints is the #epsilon values
* we try for the epsilon-constraint method..
* because of the problem being LP with linear pareto-front
* , this generates exactly %gridpoints% number of solutions
$if not set gridpoints $set gridpoints 99
$eval gridLimit 2*(%gridpoints% + 1)
Set
   g          'grid points' / g0*g%gridpoints% /
   grid(k,g)  'grid';

Parameter
   gridrhs(k,g)  'RHS of eps-constraint at grid point'
   maxg(k)       'maximum point in grid for objective'
   posg(k)       'grid position of objective'
   firstOffMax   'some counters'
   lastZero      'some counters'
*  numk(k) 'ordinal value of k starting with 1'
   numg(g)       'ordinal value of g starting with 0'
   step(k)       'step of grid points in objective functions'
   jump(k)       'jumps in the grid points traversing' ;

lastZero = 1;
loop(km1,
   numk(km1) = lastZero;
   lastZero  = lastZero +  1;
);
numg(g) = ord(g) - 1;

grid(km1,g) = yes; // Here we could define different grid intervals for different objectives
maxg(km1)    = smax(grid(km1,g), numg(g));
step(km1)    = (maxobj(km1) - minobj(km1))/maxg(km1);
gridrhs(grid(km1,g)) $(dir(km1) =  -1) = maxobj(km1) - numg(g)/maxg(km1)*(maxobj(km1) - minobj(km1));
gridrhs(grid(km1,g)) $(dir(km1) =   1) = minobj(km1) + numg(g)/maxg(km1)*(maxobj(km1) - minobj(km1));

put / ' Grid points' /;

loop(km1, put '    ', gridrhs(km1,'g0'):0:2, ' : ', step(km1):0:2, ' : ', gridrhs(km1,'g%gridpoints%'):0:2;
    put /;);
```

```
put / 'Efficient solutions'  /;

* Walk the grid points and take shortcuts if the model becomes infeasible or
* if the calculated slack variables are greater than the step size
posg(km1) = 0;
iter   = 0;
infeas = 0;
start  = jnow;

repeat
   rhs(km1) =  sum(grid(km1,g) $(numg(g) = posg(km1)), gridrhs(km1,g));
   solve mod_epsmethod maximizing a_objval using mip;
   iter = iter + 1;
   if(mod_epsmethod.modelStat<>%modelStat.optimal%,
      infeas = infeas +  1; // not optimal is in this case infeasible
      put iter:5:0, '  infeasible' /;
      lastZero =  0;
      loop(km1$(posg(km1)  > 0 and lastZero = 0), lastZero = numk(km1));
      posg(km1)$(numk(km1) <= lastZero) = maxg(km1);  // skip all solves for more demanding values of rhs(km1)
   else
      put iter:5:0;
      loop(j, put x.l(j):12:4;);
      put /;

      put '':5;
      loop(k, put z.l(k):12:2;);
      jump(km1) =  1;
*     find the first off max (obj function that hasn't reach the final grid point).
*     If this obj.fun is k then assign jump for the 1..k-th objective functions
*     The jump is calculated for the innermost objective function (km=1)
      jump(km1)$(numk(km1)  = 1) = 1 + floor(sl.L(km1)/step(km1));
      loop(km1$(jump(km1)  > 1), put '   jump';);
      put / /;
   );
*  Proceed forward in the grid
   firstOffMax =  0;
   loop(km1$(posg(km1) < maxg(km1)  and firstOffMax = 0),
      posg(km1)   =  min((posg(km1) + jump(km1)),maxg(km1));
      firstOffMax = numk(km1);
   );
   posg(km1)$(numk(km1) < firstOffMax) =  0;
   abort$(iter > %gridLimit%)  'more than %gridLimit% iterations, something seems to go wrong'  ;
until  sum(km1$(posg(km1) = maxg(km1)), 1) = card(km1) and firstOffMax =  0;

finish = jnow;
elapsed_time = (finish - start)* 60*60*24;

put /;
put 'Infeasibilities = ' , infeas:5:0 /;
put 'Elapsed time: ',elapsed_time:10:2, ' seconds' /;
```

≈ OUTPUT



Figure 1: 1st - Pareto front (100 solutions plotted in function space):



Figure 2: Pareto Solutions represented in search space

Infeasible

$$\text{Minimize} \;\; f(x_1, x_2, x_3) \; = \; 3\,x_1 - 4\,x_2 + 53\,x_3$$
$$\text{and}$$
$$\text{Minimize} \;\; g(x_1, x_2, x_3) \; = \; -10\,x_1 \; + \; x_2 - 34\,x_3$$

subject to the constraints :

$$x_1 \geqslant 0$$
$$x_2 \geqslant 0$$
$$x_3 \geqslant 0$$
$$\pi\,x_1 + x_2 - 20\,x_3 \geqslant 43$$

$$7\,x_1 + x_2 = \frac{7}{\pi} + \log_e(10) \;\approx\; 4.53075$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, pareto-optimal) solutions.

$$\mathbf{x} \textit{ does not exist.}$$

(II) Multi-Objective Non-Linear Programming Problem

Feasible

$$\textit{Maximize } f(x_1, x_2, x_3) = x_3 - \left(x_2 + \frac{x_1}{2}\right)^4 \exp\left(x_2^2 - x_1^2\right)$$

and

$$\textit{Maximize } g(x_1, x_2, x_3) = x_3 + \left(x_2 + \frac{x_1}{3}\right)^2 \exp\left(-x_3^2 \cdot x_2\right)$$

subject to the constraints :

$$-3\,x_1^2 + x_3 - 0.4 > 0$$
$$x_3 - 3.9 < 0$$
$$x_1 \cdot x_2 - x_3 = 0$$

This problem is feasible. For example, it has the following feasible solution:

$$\mathbf{x} = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 0.364008 \\ 2.1909 \\ 0.797506 \end{pmatrix}$$

Infeasible

$$\textit{Maximize } f(x_1, x_2, x_3) = x_1^2 + x_2^2 - 3\,x_1 \cdot \sin(x_2) \cdot x_3^2$$

and

$$\textit{Minimize } g(x_1, x_2, x_3) = x_1 \cdot x_2 - x_3^5 \cdot \cos(x_2)$$

subject to the constraints :

$$-x_1^2 + 3\,x_1 + x_2 - x_3 - 5.25 > 0$$
$$-3\sin(x_1 - 1) + x_2 = 0$$
$$x_3 > 0$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, pareto-optimal) solutions.

$$\mathbf{x} \textit{ does not exist.}$$

(III) Multi-Objective Mixed Integer Linear Programming Problem

Feasible

$$\begin{aligned} Minimize\ \ f(x_1,x_2,x_3) &= 5\,x_1 - x_2 + 10\,x_3 \\ &\ and \\ Maximize\ \ g(x_1,x_2,x_3) &= 13\,x_3 + 2\,x_2 - x_1 \end{aligned}$$

subject to the constraints :

$$x_3 \in \{0,1,2,\ ...,20\} = \mathbb{Z}[0,20]$$
$$x_1, x_2 \geqslant 0$$

$$15\,x_1 - x_3 \geqslant 3$$
$$x_2 + 9\,x_1 \leqslant 6$$
$$-9\,x_1 + 4x_2 = 15$$

This problem is feasible. For example, it has the following feasible solution:

$$\mathbf{x} = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 1/5 \\ 21/5 \\ 0 \end{pmatrix} = \begin{pmatrix} 0.2 \\ 4.2 \\ 0 \end{pmatrix}$$

Infeasible

$$\begin{aligned} Maximize\ \ f(x_1,x_2,x_3) &= x_1 + x_2 + 15\,x_3 \\ &\ and \\ Maximize\ \ g(x_1,x_2,x_3) &= x_1 + 2\,x_2 - 30\,x_3 \end{aligned}$$

subject to the constraints :

$$x_1,\ x_3 \in \mathbb{Z}[0,\infty)$$
$$x_2 \geqslant 0$$

$$2\,x_1 - x_2 \geqslant 6$$
$$x_2 + x_3 \geqslant 4$$
$$x_1 \leqslant 3.9$$
$$x_1 + 2\,x_2 + 3\,x_3 = 10$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, pareto-optimal) solutions.

$$\mathbf{x}\ \ does\ not\ exist.$$

(IV) Multi-Objective Mixed Integer Non-Linear Programming Problem

Feasible

$$\begin{aligned} Minimize\ \ f(x_1,x_2,x_3) &= x_1^3 \cdot x_2^3 - x_1 \cdot x_2^2 \cdot \sin(x_3) \\ &\ and \\ Maximize\ \ g(x_1,x_2,x_3) &= \sin(2x_3)\, e^{x_1 - x_2} \end{aligned}$$

subject to the constraints :

$$x_2,\ x_3 \in \mathbb{Z}[0,100]$$
$$x_1 \geqslant 0$$

$$x_1 \leqslant 100$$
$$-4\,x_1^3 + x_2 \geqslant 0$$
$$x_2 - x_3 - 3 \leqslant 0$$
$$13\,x_1 - 1.5\,x_2 + 1.98\,x_3 = 84.03226$$

This problem is feasible. For example, it has the following feasible solution:

$$\mathbf{x} = \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 2.92402 \\ 100 \\ 99 \end{pmatrix}$$

<u>Infeasible</u>

$$\textit{Maximize}\ \ f(x_1, x_2, x_3) = x_1^3 \cdot x_2^3 - x_1 \cdot x_2^2 \cdot \cos(x_3)$$
$$\textit{and}$$
$$\textit{Minimize}\ \ g(x_1, x_2, x_3) = \sin(x_1) \cdot \cos(x_2) \cdot e^{-x_3 + x_2}$$

subject to the constraints :

$$x_2,\ x_3 \in \mathbb{Z}[0, 100]$$
$$x_1 \geqslant -\pi/2$$
$$x_1 \leqslant \pi$$
$$\cos(x_1)^2 - x_2 + x_3 \geqslant 0$$
$$2 \cdot \cos(4)^2 \cdot x_1 - x_2 - 7 \cdot \cos(4)^2 = 0$$

or

$$(0.8545\,x_1 - x_2 - 2.99075 = 0)$$

This problem is infeasible due to the conflicting nature of constraints and has no feasible (and by extension, pareto-optimal) solutions.

$$\mathbf{x}\ \textit{does not exist.}$$

-------- x -------- x -------- x -------- x -------- x -------- x -------- x -------- x -------- x --------

*** The End ***